CAPÍTULO 7

( C. Castillo P. 



ARCHIVOS



Un archivo (fichero) es un conjunto de información relacionada entre sí, almacenada como una unidad en un dispositivo de almacenamiento secundario (disquete, disco  duro). Los datos almacenados en un archivo son de manera permanente de modo que pueden ser manipulados en cualquier momento. Cada archivo está referenciado por un identificador, su nombre.



CONCEPTO DE ARCHIVO BAJO WINDOWS/VISUAL BASIC

	Un archivo tiene un nombre almacenado en una carpeta junto con otros archivos de disco. Los nombres de los archivos en Windows y Visual Basic requieren de 1 a 215 caracteres (incluidos espacios en blanco), y pueden incluir también una extensión de 1 a 3 letras, normalmente son significativas y relativas al contenido del mismo. Por ejemplo:



LEAME.TXT			Archivo de texto

MISDATOS.DAT		Archivo de datos

PLANILLA.XLS		Archivo de MS-Excel

SHAKIRA.BMP		Archivo de mapa de bits

El nombre de un archivo para ser referenciado correctamente consta de:



Unidad (Especificador de unidad, por ejemplo A, B, C)

Camino (Especificador de ruta, por ejemplo \DATA\)

Nombre (Especificador de archivo, por ejemplo DEMO.DAT)



	Como ejemplo, suponga que el archivo DEMO.DAT está en la carpeta DATA de la unidad C (disco duro); la descripción completa del nombre del archivo es:



C:\DATA\DEMO.DAT



	Ahora, suponga que el archivo CONSTANTES.DAT se encuentra en la carpeta LIB, que a su vez está contenida en la carpeta DATA de la unidad C. La descripción completa del nombre de dicho archivo está dada por:



C:\DATA\LIB\CONSTANTES.DAT



	Desde el punto de vista de Visual Basic un archivo almacena los datos como un conjunto de registros, conteniendo todos ellos, generalmente, los mismos campos. Cada campo almacena un dato de tipo predefinido o de un tipo definido por el usuario. El elemento de información más simple estaría formado por un carácter.

OPERACIONES SOBRE EL SISTEMA DE ARCHIVOS

	Para manipular el sistema de archivos de un disco, Visual Basic proporciona las sentencias que a continuación se describen.



Sentencia Kill

	Permite eliminar un archivo almacenado en un disco. Su sintaxis es de la forma:



Kill NOMBRE_ARCHIVO 



Donde NOMBRE_ARCHIVO es una cadena de caracteres que identifica al archivo que se desea borrar. Se permiten caracteres comodín (* y ?). Si el archivo no existe se produce un error. Por ejemplo:



Kill “C:\TEMP\TEMPO.DAT”



La instrucción anterior elimina el archivo TEMPO.DAT ubicado en la carpeta C:\TEMP.



Sentencia Name

	Permite cambiar el nombre (renombrar) de un archivo del disco y moverlo a otro directorio si fuera preciso. Su sintaxis es la siguiente:



Name NOMBRE_ACTUAL As NOMBRE_NUEVO



Donde NOMBRE_ACTUAL es una cadena de caracteres que especifica el nombre actual del archivo y NOMBRE_NUEVO es otra cadena de caracteres que indica el nuevo nombre que se desea dar al archivo. Este nombre no debe existir, de lo contrario se obtendrá un error. Por ejemplo:



Name “C:\TEMP\TEMPO.DAT” As “C:\DATA\CURSO.DAT”



	La instrucción anterior cambia el nombre del archivo TEMPO.DAT por el nombre CURSO.DAT. El contenido del archivo permanece inalterado y su localización física se cambia de la carpeta C:\TEMP a C:\DATA.



Sentencia MkDir

Permite crear una nueva carpeta. Su sintaxis es:



MkDir NUEVA_CARPETA



Donde NUEVA_CARPETA es una cadena de caracteres que identifica la carpeta que se va ha crear. Por ejemplo:



MkDir “C:\DATA\TEMP”



Sentencia RmDir

	Elimina una carpeta vacía existente en el disco. Su sintaxis es de la forma:



RmDir NOMBRE_CARPETA



Donde NOMBRE_CARPETA es una cadena de caracteres que identifica la carpeta que se desea eliminar. Por ejemplo:



RmDir “C:\DATA\TEMP”



Sentencia ChDir 

	Permite cambiar la carpeta actual. Su sintaxis es:



ChDir NOMBRE_CARPETA



Donde NOMBRE_CARPETA es una cadena de caracteres que identifica la nueva ruta de acceso predeterminada. Por ejemplo:



ChDir “C:\DATA\GRAFICOS”

MsgBox App.Path



La propiedad Path del objeto App especifica la ruta de acceso actual.



Sentencia ChDrive

	Permite cambiar la unidad de disco actual. Su sintaxis es:



ChDrive UNIDAD



	Donde UNIDAD es un carácter que especifica la nueva unidad de disco. Si el parámetro UNIDAD es una cadena de múltiples caracteres sólo se lee la primera letra. Por ejemplo:



ChDrive “A”

OPERACIONES CON ARCHIVOS

	Para realizar alguna operación sobre un archivo hay referenciarlo mediante su nombre completo. Las operaciones básicas que se pueden realizar con los archivos son:



Abrir, preparar un archivo para hacer referencia a él.

Escribir, introducir un elemento de información a un archivo.

Leer, obtener un elemento de información de un archivo.

Modificar, alterar un elemento de información ya existente en un archivo.

Cerrar, evitar cualquier otra referencia al archivo en tanto no se le abra otra vez.



TIPOS DE ARCHIVOS

	Los tipos de archivos dependen del modo como están organizados los registros y de la forma de accesar a la los datos contenidos en ellos. En Visual Basic existen tres tipos de archivos de datos, estos son:



Archivos secuenciales (acceso secuencial).

Archivos aleatorios (acceso aleatorio).

Archivos binarios (acceso binario).



	A continuación pasamos a describir cada uno de ellos en forma detallada.

ARCHIVOS DE ACCESO SECUENCIAL

	En un archivo de acceso secuencial los registros se almacenan siguiendo una a otro, según el orden en que son ingresados. Cuando se lee la información, se empieza por el primer registro y se continua al siguiente hasta alcanzar el final. Las sentencias y funciones necesarias para manipular archivos de tipo secuencial se presentan a continuación.



Sentencia Open

	Permite abrir un archivo. La sintaxis para esta sentencia es la siguiente:



Open NOMBRE_ARCHIVO For MODO As # NÚMERO_ARCHIVO



	Donde NOMBRE_ARCHIVO es una cadena que especifica el nombre del archivo que se debe ser abierto en MODO Output, Append o Input.



Modo�Descripción��Output�Escritura de datos. Si el archivo existe, su contenido actual se destruye. Si el archivo no existe, se crea.��Append�Añadir datos. Los datos son añadidos a partir de los últimos existentes. Si el archivo no existe, se crea.��Input�Lectura de datos. La lectura empieza desde el principio del archivo. Si el archivo no existe, se produce un error.��	El parámetro NÚMERO_ARCHIVO es un entero cuyo valor debe estar comprendido entre 1 y 511. Este número será asociado al nombre del archivo mientras éste permanezca abierto. Para obtener el número del siguiente archivo disponible se utiliza la función FreeFile().

Como ejemplo suponga que se requiere abrir el archivo DEMO.DAT ubicado en la carpeta C:\DATA, la instrucción sería la siguiente:



Dim N1 As Integer

N1 = FreeFile()

Open “C:\DATA\DEMO.DAT” For Output As # N1



Sentencia Print

	Permite escribir datos secuencialmente en un archivo. Su sintaxis es:



Print # NÚMERO_ARCHIVO, LISTA_DE_EXPRESIONES



	Donde NÚMERO_ARCHIVO es el número utilizado cuando el archivo fue abierto. LISTA_DE_EXPRESIONES es un conjunto de expresiones (numéricas, de cadena, de fecha, etc.) separadas por punto y coma (;) que serán escritas en el archivo.

	La sentencia Print escribe en el archivo una imagen de los datos tal y como se habrían visualizado sobre el formulario con la sentencia Print. Por ejemplo:

Dim N1 As Integer

N1 = FreeFile()

Open “C:\DATA\DEMO.DAT” For Output As # N1

Print # N1, “Visual Basic es fácil”; “, ”; Date()



	Al ejecutarse el código anterior se escribiría en el archivo la siguiente información:



Visual Basic es fácil, 24/04/2001



	Como se observa, al utilizar la sentencia Print se deben delimitar los datos para que se impriman correctamente.



Sentencia Write

	Permite escribir datos secuencialmente en un archivo. Su sintaxis es:



Write # NÚMERO_ARCHIVO, LISTA_DE_EXPRESIONES



	Donde NÚMERO_ARCHIVO es el número utilizado cuando el archivo fue abierto. LISTA_DE_EXPRESIONES es un conjunto de expresiones (numéricas, de cadena, de fecha, etc.) separadas por punto y coma (;) que serán escritas en el archivo.

La sentencia Write inserta comas (,) entre las expresiones de la LISTA_DE_EXPRESIONES, por tanto no es necesario poner delimitadores explícitamente como en el caso de la sentencia Print. 

Cuando se utiliza la sentencia Write para escribir información en un archivo, se siguen distintas convenciones universales, de modo que los datos siempre se pueden leer e interpretar correctamente, independientemente de la configuración regional, estas convenciones son:



Los datos numéricos siempre se escriben utilizando la coma (,) como separador decimal.

Para datos de tipo Boolean se imprime # TRUE # o # FALSE #.

Los datos de tipo Date se escriben en el archivo usando el formato de fecha universal (fechas como # aaaa-mm-dd # y horas como # hh:mm:ss #).



A manera de ejemplo considere el siguiente segmento de código:



Dim N1 As Integer

N1 = FreeFile()

Open “C:\DATA\DEMO.DAT” For Output As # N1

Write # N1, “Visual Basic es fácil”; Date()



	La ejecución de este código escribiría en el archivo la siguiente información:



Visual Basic es fácil", # 2002-04-11 #



Sentencia Close

	Cierra uno archivo abierto mediante la sentencia Open. Su sintaxis es la siguiente:



Close # NÚMERO_ARCHIVO [, # NÚMERO_ARCHIVO, . . .]



Donde NÚMERO_ARCHIVO es el número con el cual se abrió el archivo. Por ejemplo:



Close # 1, # 2 



La instrucción anterior cierra los archivos asociados con los números 1 y 2. La siguiente sentencia cierra todos los archivos abiertos.



Close



Sentencia Input

	Permite leer datos de un archivo secuencial y los asigna a las variables especificadas. Su sintaxis es:



Input # NÚMERO_ARCHIVO, VARIABLE1 [, VARIABLE2, . . .]



	Donde NÚMERO_ARCHIVO es el número utilizado cuando el archivo fue abierto. VARIABLE1, VARIABLE2, . . . son los nombres de las variables que han de recibir los correspondientes datos del archivo.

Los datos del archivo deben aparecer en el mismo orden que tienen las variables en la sentencia Input y deben coincidir con variables del mismo tipo de datos. Por ejemplo: 



Dim N1 As Integer

N1 = FreeFile()

Open “C:\DATA\DEMO.DAT” For Input As # N1

Dim A As Integer

Dim B As Double

Dim S As String

Dim F As Date

Input # N1, A, B, S, F



El segmento de código anterior espera encontrar en el archivo un entero, un real, una cadena y una fecha, en ese orden (separados por comas o un retorno de carro).



Sentencia Line Input

	Permite leer una línea de un archivo secuencial ignorando los delimitadores (comas) y la asigna a una variable tipo cadena. Su sintaxis es:



Line Input # NÚMERO_ARCHIVO, VARIABLE



	Donde NÚMERO_ARCHIVO es el número utilizado cuando el archivo fue abierto. VARIABLE es el nombre de una variable tipo cadena de caracteres.

	La sentencia Line Input se utiliza especialmente para leer un archivo de texto línea a línea, ya que esta sentencia lee todos los caracteres del archivo hasta que encuentra un retorno de carro, entonces continua en la siguiente línea y así sucesivamente. Por ejemplo:



Dim N1 As Integer, LINE1 As String, LINE2 As String

N1 = FreeFile()

Open “C:\DATA\DEMO.TXT” For Output As # N1

Print # N1, “Línea de prueba 1”

Print # N1, “Línea de prueba 2”

Close # N1

Open “C:\DATA\DEMO.TXT” For Input As # N1

Line Input # N1, LINE1

MsgBox LINE1

Line Input # N1, LINE2

MsgBox LINE2



La ejecución del código anterior produce la siguiente salida:



� INCRUSTAR PBrush  ���

Función Input

	Retorna los siguientes � INCRUSTAR Equation.3  ��� caracteres de un archivo secuencial y los asigna a una variable de cadena. Su sintaxis es de la forma:



VARIABLE = Input(� INCRUSTAR Equation.3  ���, # NÚMERO_ARCHIVO)



	A diferencia de la sentencia Input, la función Input() retorna todos los caracteres que lee, incluyendo comas, retornos de carro, continuaciones de línea, etc. Por ejemplo:



Dim N1 As Integer, S As String

N1 = FreeFile()

Open “C:\DATA\DEMO.TXT” For Output As # N1

Print # N1, “Línea de prueba 1” 

Print # N1, “Línea de prueba 2”

Close # N1

Open “C:\DATA\DEMO.TXT” For Input As # N1

S = Input(24, # N1)

MsgBox S



La ejecución del código anterior produce la siguiente salida:



� INCRUSTAR PBrush  ���

Función EOF

	Especifica si se ha llegado al final de un archivo. Su sintaxis es de la forma:



VARIABLE = EOF(NÚMERO_ARCHIVO)



	Se utiliza EOF() para evitar producir un error al intentar obtener información más allá del final del archivo. EOF() retorna un valor True si se ha alcanzado el final del archivo y False en caso contrario. Por ejemplo:



Dim N1 As Integer

Dim CADENA As String

N1 = FreeFile()

Open “C:\DATA\DEMO.TXT” For Input As # N1

While Not EOF(N1)

	Line Input # N1, CADENA

	Print CADENA

	Wend

	Close # N1



	Este segmento de código lee y visualiza cada línea del archivo de texto DEMO.TXT. El bucle finaliza cuando se detecta el final del archivo. Para que el código anterior funcione correctamente, no olvide poner la propiedad AutoRedraw del formulario a True.

	Como ejemplo final del uso de archivos secuenciales vamos a desarrollar un sencillo editor de texto. Este editor aunque sus prestaciones son bastante limitadas va ha servir para poner en práctica lo aprendido recientemente. 



� INCRUSTAR PBrush  ���



	El menú que se muestra en la figura obedece a la siguiente descripción:



Caption�Name�ShortCut��&Archivo�MnuArchivo�Ninguno��. . . &Nuevo�MnuArchivoNuevo�Ninguno��. . . &Abrir�MnuArchivoAbrir�Ninguno��. . . &Guardar�MnuArchivoGuardar�Ninguno��. . . -�MnuArchivoLinea�Ninguno��. . . &Salir�MnuArchivoSalir�Ctrl + X��Luego proceda a ubicar los siguientes controles sobre el formulario:



	1 cuadro de texto

	1 control CommonDialog



	Seguidamente debe establecer las propiedades que se indican:



Form1

�Nombre�FrmEditor���Caption�Editor��

Text1

�Nombre�TxtEditor���MultiLine�True���ScrollBars�3-Both���Text���

CommonDialog1

�Nombre�CommonDialog1���CancelError�True��

	Una vez establecidas las propiedades de la interfaz ingresar el código que se muestra:



Private Sub Form_Resize()

    TxtEditor.Move 0, 0, ScaleWidth, ScaleHeight

End Sub

Private Sub MnuArchivoNuevo_Click()

TxtEditor = “”

End Sub



Private Sub MnuArchivoAbrir_Click()

Dim FILTRO As String, FILE As String

Dim N1 As Integer, CADENA As String

On Error GoTo ERROR_ABRIR

FILTRO = “Archivos de texto (*.TXT)|*.TXT”

CommonDialog1.Filter = FILTRO

CommonDialog1.ShowOpen

FILE = CommonDialog1.FileName

N1 = FreeFile()

CADENA = “”

TxtEditor = “”

Open FILE For Input As # N1

While Not EOF(N1)

Input # N1, CADENA

TxtEditor = TxtEditor & CADENA & vbCrLf

Wend

Close # N1

FrmEditor.Caption = “Editor - ” & FILE

SALIR_ABRIR:

    Exit Sub

ERROR_ABRIR:

    MsgBox Err.Description

    Resume SALIR_ABRIR

End Sub

Private Sub MnuArchivoGuardar_Click()

Dim FILTRO As String, FILE As String

Dim N1 As Integer

On Error GoTo ERROR_GUARDAR

FILTRO = “Archivos de texto (*.TXT)|*.TXT”

CommonDialog1.Filter = FILTRO

CommonDialog1.ShowSave

FILE = CommonDialog1.FileName

N1 = FreeFile()

Open FILE For Output As # N1

Print # N1, TxtEditor

Close # N1

FrmEditor.Caption = “Editor - ” & FILE

SALIR_GUARDAR:

Exit Sub

ERROR_GUARDAR:

MsgBox Err.Description

Resume SALIR_GUARDAR

End Sub



	La sentencia Err.Description retorna una cadena que contiene la descripción asociada a un error en tiempo de ejecución.

	La sentencia Resume continua la ejecución en la etiqueta especificada cuando termina una rutina de gestión de errores. 

ARCHIVOS DE ACCESO ALEATORIO

	En los archivo de acceso aleatorio el almacenamiento de los datos se hace mediante registros (todos de la misma longitud), lo cuales son identificados mediante un único número denominado índice. El primer registro de un archivo tiene como índice 1, el segundo tiene índice 2 y así sucesivamente. La información contenida en un archivo de este tipo puede ser accedida en cualquier secuencia, ya que cada registro individual se asocia con su respectivo índice y puede ser leído, escrito o actualizado.

Las sentencias y funciones necesarias para manipular archivos de tipo aleatorio se presentan a continuación.



Sentencia Open

	Permite abrir un archivo. La sintaxis para acceder aleatoriamente a un archivo es:



Open NOMBRE_ARCHIVO For Random As # NÚMERO_ARCHIVO Len = LON_REG



	Donde NOMBRE_ARCHIVO es una cadena que especifica el nombre del archivo que se debe ser abierto en modo Random.

El parámetro NÚMERO_ARCHIVO es un entero cuyo valor está comprendido entre 1 y 511. Este número será asociado con el nombre del archivo mientras permanezca abierto.

LON_REG es un entero que establece la longitud del registro para archivos aleatorios.

Sentencia Put

	Permite grabar un registro en un archivo abierto para acceso aleatorio. Su sintaxis es:



Put # NÚMERO_ARCHIVO, NÚMERO_REG, VARIABLE



	Donde NÚMERO_ARCHIVO es el número bajo el cual se abrió el archivo, NÚMERO_REG es el número correspondiente al registro que se va ha grabar y VARIABLE contiene los datos a escribir en el archivo. Por ejemplo:



Dim N1 As Integer, REG As DISTRITO

REG.ID_DISTRITO = “L09”

REG.NOMBRE = “Chorrillos”

N1 = FreeFile()

Open “C:\DATA\RAND1.DAT” For Random As # N1 Len = Len(REG)

Put # N1, 1, REG

Close # N1



	El segmento de código anterior utiliza una variable REG de tipo DISTRITO, cuya definición es la siguiente:



Private Type DISTRITO

ID_DISTRITO As String * 3

NOMBRE As String * 30

End Type

Sentencia Get

	Permite leer un registro procedente de un archivo de acceso aleatorio, almacenando los datos en una variable específica. Su sintaxis es de la forma:



Get # NÚMERO_ARCHIVO, NÚMERO_REG, VARIABLE



	Donde NÚMERO_ARCHIVO es el número bajo el cual se abrió el archivo, NÚMERO_REG es el número correspondiente al registro que se va ha leer y VARIABLE almacena los datos del registro leído. Por ejemplo: 



Dim N1 As Integer, I As Integer, REG As DISTRITO

N1 = FreeFile()

Open “C:\DATA\RAND1.DAT” For Random As # N1 Len = Len(REG)

I = 1

While Not EOF(# N1)

Get # N1, I, REG

MsgBox REG.ID_DISTRITO & “ ” & REG.NOMBRE

I = I + 1

	Wend

Close # N1



	Cuando EOF() se utiliza con un archivo aleatorio, retorna un valor True si una sentencia Get intenta leer y no puede porque ha alcanzado el final del archivo.

Función LOF

	Retorna el número de bytes (caracteres) que ocupa un determinado archivo abierto mediante la sentencia Open. Su sintaxis es:



VARIABLE = LOF(# NÚMERO_ARCHIVO)



	Donde NÚMERO_ARCHIVO es el número con el que se abrió el archivo.

	Esta función es de utilidad, porque aplicada a un archivo de acceso aleatorio, permite conocer el número de registros almacenados en el archivo. Para ello debe dividir el valor retornado entre la longitud del registro. Como ejemplo, considere lo siguiente:



Dim N1 As Integer, REG As DISTRITO

Dim NUM_REGS As Integer, I As Integer

N1 = FreeFile()

Open “C:\DATA\RAND1.DAT” For Random As # N1 Len = Len(REG)

NUM_REGS = LOF(N1) / Len(REG)

For I = 1 To NUM_REGS

Get # N1, I, REG

MsgBox REG.ID_DISTRITO & “ ” & REG.NOMBRE

Next

Close # N1

Función Loc

	Esta función retorna la posición actual dentro de un fichero. Su sintaxis es:



VARIABLE = Loc(# NÚMERO_ARCHIVO)



	La función Loc() aplicada a un archivo de acceso aleatorio retorna el número del último registro leído o grabado en el archivo especificado. Por ejemplo:



Dim N1 As Integer, REG As DISTRITO

Dim NUM_REGS As Integer, I As Integer

N1 = FreeFile()

Open “C:\DATA\RAND1.DAT” For Random As # N1 Len = Len(REG)

NUM_REGS = LOF(N1) / Len(REG)

I = 1

Do While True

Get # N1, I, REG

MsgBox REG.ID_DISTRITO & “ ” & REG.NOMBRE

I = I + 1

If Loc(N1) = NUM_REGS Then Exit Do

Loop

Close # N1



La sentencia If finaliza el bucle si se ha alcanzado el último registro.

Como ejemplo final vamos a desarrollar una aplicación que permita realizar el mantenimiento de los datos almacenados en el archivo C:\DATA\CURSO.DAT.



� INCRUSTAR PBrush  ���



Los botones ubicados en el marco Navegador (Primero, Anterior, Siguiente y Ultimo, de izquierda a derecha) permiten desplazarse a través de los registros del archivo. Los botones ubicados en el marco Mantenimiento (Nuevo, Editar, Guardar y Eliminar, de izquierda a derecha) permiten realizar las operaciones básicas de mantenimiento de los registros.

	Para el desarrollo de la presente aplicación ubicar los siguientes controles al formulario:



	3 marcos

	4 etiquetas

	4 cuadros de texto

	9 botones de comando

Seguidamente proceda a establecer las propiedades que se indican:



Form1

�Nombre�FrmCurso���Caption�Mantenimiento de cursos���BorderStyle�3-Fixed Dialog���StarUpPosition�2-CenterScreen��

Frame1

�Nombre�FraIngreso���Caption���

Frame2

�Nombre�FraNavegador���Caption�Navegador��

Frame3

�Nombre�FraMantenimiento���Caption�Mantenimiento��

Label1

�Nombre�LblCodigo���Caption�Código:��

Label2

�Nombre�LblNombre���Caption�Nombre:��

Label3

�Nombre�LblVacantes���Caption�Vacantes:��

Label4

�Nombre�LblProfesor���Caption�Profesor:��

Text1

�Nombre�TxtCodigo���Locked�True���Text���

Text2

�Nombre�TxtNombre���Locked�True���Text���

Text3

�Nombre�TxtVacantes���Locked�True���Text���

Text4

�Nombre�TxtProfesor���Locked�True���Text���

Command1

�Nombre�CmdPrimero���Caption����Picture�C:\FundVB\Bitmaps\First.bmp���Style�1-Graphical��



Command2

�Nombre�CmdAnterior���Caption����Picture�C:\FundVB\Bitmaps\Previous.bmp���Style�1-Graphical��

Command3

�Nombre�CmdSiguiente���Caption����Picture�C:\FundVB\Bitmaps\Next.bmp���Style�1-Graphical��

Command4

�Nombre�CmdUltimo���Caption����Picture�C:\FundVB\Bitmaps\Last.bmp���Style�1-Graphical��

Command5

�Nombre�CmdNuevo���Caption����Picture�C:\FundVB\Bitmaps\New.bmp���Style�1-Graphical��

Command6

�Nombre�CmdEditar���Caption����Picture�C:\FundVB\Bitmaps\Edit.bmp���Style�1-Graphical��



Command7

�Nombre�CmdGuardar���Caption����Picture�C:\FundVB\Bitmaps\Save.bmp���Style�1-Graphical��

Command8

�Nombre�CmdEliminar���Caption����Picture�C:\FundVB\Bitmaps\Delete.bmp���Style�1-Graphical��

Command9

�Nombre�CmdSalir���Caption�&Salir���Picture�C:\FundVB\Bitmaps\Exit.bmp���Style�1-Graphical��

Luego, proceda a transcribir el código mostrado a continuación:



Private Type CURSO

ID_CURSO As String * 3

NOMBRE As String * 30

VACANTES As Integer

PROFESOR As String * 25

ESTADO As Boolean

End Type



Dim N1 As Integer, POSICION As Integer

Dim REG As CURSO

Private Sub MODO_EDITAR(ByVal Ok As Boolean)

TxtCodigo.Locked = Not Ok

TxtNombre.Locked = Not Ok

TxtVacantes.Locked = Not Ok

TxtProfesor.Locked = Not Ok

CmdNuevo.Enabled = Not Ok

CmdEditar.Enabled = Not Ok

CmdGuardar.Enabled = Ok

CmdEliminar.Enabled = Not Ok

CmdPrimero.SetFocus

If Ok Then TxtCodigo.SetFocus

End Sub



Private Sub ABRIR_ARCHIVO()

On Error GoTo ERROR_ABRIR:

N1 = FreeFile()

Open “C:\DATA\CURSO.DAT” For Random As # N1 Len = Len(REG)

SALIR_ABRIR:

Exit Sub

ERROR_ABRIR:

MsgBox Err.Description

Resume SALIR_ABRIR:

End Sub



Private Sub Form_Load()

Call ABRIR_ARCHIVO

End Sub

Private Sub Form_Activate()

MODO_EDITAR False

Call CmdPrimero_Click

End Sub



Private Sub CmdPrimero_Click()

POSICION = 1

Get # N1, POSICION, REG

TxtCodigo = REG.ID_CURSO

TxtNombre = REG.NOMBRE

TxtVacantes = Str(REG.VACANTES)

TxtProfesor = REG.PROFESOR

End Sub



Private Sub CmdAnterior_Click()

POSICION = Loc(N1) - 1

If POSICION = 0 Then

Get # N1, 1, REG

MsgBox “Estamos en el primer registro”

Else

Get # N1, POSICION, REG

End If

TxtCodigo = REG.ID_CURSO

TxtNombre = REG.NOMBRE

TxtVacantes = Str(REG.VACANTES)

TxtProfesor = REG.PROFESOR

End Sub

Private Sub CmdSiguiente_Click()

Dim ULTIMO As Integer

ULTIMO = LOF(N1) / Len(REG)

POSICION = Loc(N1) + 1

If POSICION = ULTIMO + 1 Then

Get # N1, ULTIMO, REG

MsgBox “Estamos en el último registro”

Else

Get # N1, POSICION, REG

End If

TxtCodigo = REG.ID_CURSO

TxtNombre = REG.NOMBRE

TxtVacantes = Str(REG.VACANTES)

TxtProfesor = REG.PROFESOR

End Sub



Private Sub CmdUltimo_Click()

POSICION = LOF(N1) / Len(REG)

If POSICION <> 0 Then

Get #N1, POSICION, REG

TxtCodigo = REG.ID_CURSO

TxtNombre = REG.NOMBRE

TxtVacantes = Str(REG.VACANTES)

TxtProfesor = REG.PROFESOR

End If

End Sub

Private Sub CmdNuevo_Click()

POSICION = LOF(N1) / Len(REG) + 1

MODO_EDITAR True

TxtCodigo = “”

TxtNombre = “”

TxtVacantes = “”

TxtProfesor = “”

TxtCodigo.SetFocus

End Sub



Private Sub CmdEditar_Click()

MODO_EDITAR True

End Sub



Private Sub CmdGuardar_Click()

REG.ID_CURSO = Trim(TxtCodigo)

REG.NOMBRE = Trim(TxtNombre)

REG.VACANTES = TxtVacantes

REG.PROFESOR = Trim(TxtProfesor)

REG.ESTADO = True

Put # N1, POSICION, REG

MODO_EDITAR False

End Sub



Private Sub CmdEliminar_Click()

Dim N2 As Integer

Dim I As Integer, J As Integer

REG.ESTADO = False

Put # N1, POSICION, REG

N2 = FreeFile()

Open “C:\DATA\TEMPO.DAT” For Random As # N2 Len = Len(REG)

I = 1

J = 1

While Not EOF(N1)

Get # N1, I, REG

If REG.ESTADO Then

Put # N2, J, REG

J = J + 1

End If

I = I + 1

Wend

Close # N1, # N2

Kill “C:\DATA\CURSO.DAT”

Name “C:\DATA\TEMPO.DAT” As “C:\DATA\CURSO.DAT”

Call ABRIR_ARCHIVO

Call CmdPrimero_Click

End Sub



Private Sub CmdSalir_Click()

	Close # N1

End

End Sub

ARCHIVOS DE ACCESO BINARIO

Un archivo binario contiene más que simplemente texto. Puede contener imágenes, sonido, hojas de cálculo, o documentos concebidos para el procesamiento de texto.

El acceso binario permite la posibilidad de tratar cualquier archivo como una secuencia numerada de bytes, independientemente de la estructura del mismo. Los bytes ocupan las posiciones 1, 2, 3, etc. Por ejemplo, si se requiere recuperar un dato de tipo entero (Integer, 2 bytes) de la posición 3 del archivo, serían recuperados los bytes 3 y 4 para poder formar el valor del entero. Por tanto, antes de trabajar con archivos binarios es necesario conocer cómo fueron escritos los datos que contiene para poder recuperarlos correctamente.



Sentencia Open

	Permite abrir un archivo para acceso binario. Su sintaxis es de la forma:



Open NOMBRE_ARCHIVO For Binary As # NÚMERO_ARCHIVO



	Donde NOMBRE_ARCHIVO es una cadena que especifica el nombre del archivo que se debe ser abierto en modo Binary.

El parámetro NÚMERO_ARCHIVO es un entero cuyo valor está comprendido entre 1 y 511. Este número será asociado con el nombre del archivo mientras permanezca abierto.



Sentencia Put

	Permite grabar en un archivo binario tantos bytes como haya en una variable. Su sintaxis es:



Put # NÚMERO_ARCHIVO, POSICIÓN, VARIABLE



	Donde NÚMERO_ARCHIVO es el número bajo el cual se abrió el archivo. POSICIÓN es el número de byte a partir del cual se han de grabar los datos contenidos en VARIABLE.



Sentencia Get

	Permite leer de un archivo binario tantos bytes como quepan en una variable. Su sintaxis es de la forma:



Get # NÚMERO_ARCHIVO, POSICIÓN, VARIABLE



	Donde NÚMERO_ARCHIVO es el número bajo el cual se abrió el archivo. POSICIÓN es el número de byte a partir del cual se han de leer los datos almacenados en VARIABLE.



Sentencia Seek

	Permite situar la posición de lectura o de escritura en una posición determinada dentro del archivo. Su sintaxis es:



Get # NÚMERO_ARCHIVO, POSICIÓN, VARIABLE



	Donde POSICIÓN es el número de byte a partir del cual queremos leer o escribir dentro del archivo.

Como ejemplo vamos a desarrollar una aplicación que permite guardar “cifrados” los textos ingresados en un cuadro de texto. Para ello crear un nuevo proyecto y ubicar en el formulario un cuadro de texto (TxtMensaje) y un botón de comando (CmdCifrar). Luego, ingrese el siguiente código:



Const CLAVE As Integer = 3

Private Sub CmdCifrar_Click()

Dim N1 As Integer, I As Integer

Dim CAR As String * 1

N1 = FreeFile()

Open “C:\DATA\DEMO.BIN” For Binary As # N1

For I = 1 To Len(TxtMensaje)

  	CAR = Chr((Asc(Mid(TxtMensaje, I, 1)) + CLAVE) Mod 256)

Put # N1, , CAR

Next

Close # N1

End Sub



	En el código la función Mid() obtiene el carácter “I” de la caja de texto, la función Asc() obtiene su código ANSI, al que sumamos el valor de CLAVE, para después obtener el resto de la división entre 256, con el fin de mantenernos en el rango de 0 a 255 (rango de valores de la tabla de caracteres ANSI). Por último, la función Chr() retorna el carácter correspondiente al valor obtenido, el cual es almacenado en el archivo binario. 

Por ejemplo si ingresa el mensaje “HOLA” se almacena en el archivo como “KROD” (lo puede comprobar mediante el Bloc de notas), ya que el valor ANSI de la “H” es 72, este carácter al sumarle el valor de CLAVE sería el 75 (72 + 3), que es la “K”, y así sucesivamente (ver la tabla de caracteres ANSI).

El descifrado sería el proceso inverso, para ello crear un nuevo formulario y ubicar un botón de comando (CmdDescifrar), luego ingresar el código siguiente:



Const CLAVE As Integer = 3

Private Sub CmdDescifrar_Click()

Dim N1 As Integer, I As Integer

Dim CAR As String * 1, CADENA As String

N1 = FreeFile()

Open “C:\DATA\DEMO.BIN” For Binary As # N1

Get # N1, , CAR

While Not EOF(N1)

CAR = Chr((Asc(CAR) + (256 - Val(CLAVE))) Mod 256)

CADENA = CADENA & CAR

Get # N1, , CAR

Wend

MsgBox CADENA

End Sub

